 A sum of binomially distributed variables
The idea of sums of random variables is very important and the applications are very com​mon. For some variables, e.g. the normal distribution, the distribution of a sum is straight​forward. If we sum three normally distributed variables the sum is again a normal distri​bution.

The binomial distribution is extremely common and rather straight forward: we look at n items (people, things, events, etc) and count the number of faulty items (number of men, number of bad things, number of incorrect events, etc). This number is our measurement. There are two more features needed: 

1. independence between the events

2. the fault rate (p) must be equal all the time

Of course, it might happen that e.g. the fault rate changes over time and this is of course one thing we want to discover by using the binomial distribution as a random model.

For more information of sums of variables we suggest chapter 4 in the document 'A course in statistics' and a number of so-called computer macros (e.g. %LinC, %CLT, %Die) described in the document 'A collection of diagrams'. A thorough discussion of the binomial distribution is found in chapter 8 of 'A course in statistics' and by macros (e.g. %Bpdfcdf, %Bhist, %Bdata, %MultDist, etc) in 'A collection of diagrams'.

Example I

As an example we will use a PCA (Printed Circuit Assembly) equipped as follows (the 'fault rate' is the same as the probability of a component being 'not OK'):


15 
components of type A
(fault rate pA = 0.03)


20
components of type B
(fault rate pB = 0.05)


25
components of type C
(fault rate pC = 0.08)


4
connectors of type D
(fault rate pD = 0.10)

3
labels of type E
(fault rate pE = 0.12)

(some schools will say that we have 15 + 20 + 25 + 4 + 3 = 67 opportunities for fault.)

We will regard each one of the components as either 'OK' or 'not OK'. This goes also for the con​nectors and the labels. E.g. a label is regarded as 'OK' when it contains the correct information, is read​able and correctly placed. Otherwise it is 'not OK'. All components, connectors, etc fail inde​pendently of one another. We will designate the sum of incorrect components etc as Y.

The model.  Now we can write Y as the following model where each term is either '0' or '1' if the outcome is 'OK' or 'not OK' respectively:

The expected value.  First we will calculate the expected value (theoretical mean) of Y, i.e. E(Y). In general, the expected value of a discrete variable, say X, is calculated using the following ex​pres​sion:

    (”the value xi multiplied with the probability of that value”)

This becomes for e.g. A1:

[image: image3.wmf]A

p

p

X

P

X

P

A

E

A

=

=

×

+

=

×

=

4

3

4

2

1

4

3

4

2

1

)

1

(

1

)

0

(

0

)

(

0

1


Now we will apply this to the whole expression above together with the theory of linear combinations of variables (where we sum the expected values):

This can be simplified as  
[image: image5.wmf]p

n

n

n

n

n

Y

E

E

D

C

B

A

×

+

+

+

+

=

)

(

)

(

  just because of the general possi​bility of writing a sum as  

The standard deviation.  Using the model above we can calculate the standard deviation of Y. We do that using the fact that the variance (the standard deviation squared) is additive. For each one of the terms above the variance is as follows (this can easily be determined from the definition of the variance):

This gives us the following expression for the variance of Y:

This expression can also be simplified:

Simulation of example I

In order to show the results above, we will calculate the expected value and the standard deviation and compare this to a simulated result.

 let c1(1) = 15                # Number of A-components.

 let c1(2) = 20                # Number of B-components.

 let c1(3) = 25                # Number of C-components.

 let c1(4) = 4                 # Number of D-connectors.

 let c1(5) = 3                 # Number of E-labels.

 set c2

 15(0.03) 20(0.05) 25(0.08)    # 15 'pA', 20 'pB', 25 'pC' etc.

 4(0.10)  3(0.12)

 end

 let k1 = sum(c1) * mean(c2)   # The expected value.

 let k2 = sqrt(sum(c1) * (mean(c2)*(1-mean(c2)) - stand(c2)**2)) # Stand.

 # Simulating ---------------------------------------------------------

 let k3 = 10000                # Number of simulated PCA's.

 name c1 'A1' c16 'B1' c36 'C1' c61 'D1' c65 'E1'  # Names some columns.

 random k3 c1-c15;             # Creates 0's and 1's in c1-c15 with the

 bernoulli 0.03.               # probability of 0.03 of a ”1”.

 random k3 c16-c35;            # Creates 0's and 1's in c16-c35 with the

 bernoulli 0.05.               # probability of 0.05 of a ”1”.

 random k3 c36-c60;            # Creates 0's and 1's in c36-c60 with the

 bernoulli 0.08.               # probability of 0.08 of a ”1”.

 random k3 c61-c64;            # Creates 0's and 1's in c61-c64 with the

 bernoulli 0.10.               # probability of 0.10 of a ”1”.

 random k3 c65-c67;            # Creates 0's and 1's in c65-c67 with the

 bernoulli 0.12.               # probability of 0.12 of a ”1”.

 rsum c1-c67 c71               # Each row of c70 is the result of 1 PCA.

 describe c71                  # Describes the result.
Conclusion.  There is a good agreement between the theoretical mean and standard deviation, stored in k1 and k2, and the mean and standard deviation in the printout.

Calculating mean value and standard deviation of the distribution of Y for example I

The theory of calculating the probability distribution is very complicated and involves advanced methods (see appendix for some more details). However, we will use the probabilities that were calculated in the appendix I below when simulating some results using the macro %CreDist:

The probability of 0 to 14:

   0.012623   0.057359   0.128022   0.187090   0.201347   0.170172

   0.117623   0.068372   0.034109   0.014832   0.005690   0.001945

   0.000597   0.000166   0.000042

Do the following steps:

1. Run %CreDist once

2. Copy the following data into c1 and c2:
    0 0.012623              # 0 fault on the PCA, probability 0.012623.
    1 0.057359              # 1 fault on the PCA, probability 0.057359.
    2 0.128022              # Etc.
    3 0.187090
    4 0.201347
    5 0.170172
    6 0.117623
    7 0.068372
    8 0.034109
    9 0.014832
   10 0.005690
   11 0.001945
   12 0.000597
   13 0.000166
   14 0.000042

3. Enter '2000' in cell 15 of c1.

4. Rerun the macro %CreDist

When running this input the printout will show exactly the same theoretical mean and standard deviation as calculated above. If we know the distribution of a variable we can always calculate the theoretical mean and standard deviation using the proper formulas (not shown in this document). 

Earlier we also managed to reformulate our problem and see that it was a linear combination of more simple variables. Thereby we managed to calculate the theoretical mean and standard deviation without knowing the distribution. 

However, without the distribution we cannot answer questions such as 'what is the probability of getting a value larger than 11?'.

Comparing the theoretical distribution and the simulated histogram.  Finally we will compare a histogram over simulated data with the theoretically derived distribution. We use the same com​mands as above but simulate 100 000 PCA:

# Simulating ---------------------------------------------------------

 Let k3 = 100000               # Number of simulated PCA's.

 name c1 'A1' c16 'B1' c36 'C1' c61 'D1' c65 'E1'  # Names some columns.

 random k3 c1-c15;             # Creates 0's and 1's in c1-c15 with the

 bernoulli 0.03.               # probability of 0.03 of a ”1”.

 random k3 c16-c35;            # Creates 0's and 1's in c16-c35 with the

 bernoulli 0.05.               # probability of 0.05 of a ”1”.

 random k3 c36-c60;            # Creates 0's and 1's in c36-c60 with the

 bernoulli 0.08.               # probability of 0.08 of a ”1”.

 random k3 c61-c64;            # Creates 0's and 1's in c61-c64 with the

 bernoulli 0.10.               # probability of 0.10 of a ”1”.

 random k3 c65-c67;            # Creates 0's and 1's in c65-c67 with the

 bernoulli 0.12.               # probability of 0.12 of a ”1”.

 rsum c1-c67 c71               # Each row of c70 is the result of 1 PCA.

 tally c71;                    # Gives number of 0's, 1's etc in c71.

 store c72 c73.                # Stores the result in c72 and c73.

let c73 = c73/k3               # Gives the simulated proportions.

Compare the simulated contents of c72 and c73 with the theoretical input to the macro. There should be a rather good agreement between the results.

Example II

As the second example we use a common situation namely where we have a number of mea​sure​ments on an item. Suppose e.g. that we have 10 different continuous variables on a piece of equip​ment and the requirements states that all values must be within the specifications in order to approve the equipment.

For the sake of the example we simulate data according to the table below. (See the %GamArea macro for examples of gamma variables.)

	Columns
	Distributions
	Parameters
	LSL
	USL

	c1 – c4
	Normal
	[50; 5]
	41
	62

	c5 – c7
	Gamma
	[2.5; 0.9]
	0.5
	5.4

	c8 – c10
	Gamma
	[15.1; 2.3]
	3.1
	9.2


As we now have a sample of 10 measurements per item it would tempting to say that we have a binomial distribution with ‘n' = 10 and a certain ‘p'. However, this is not true because the fault rate is most likely different for the variables and thus we would violate one of the requirements for the binomial distribution.

In these kinds of situations we most likely are interested in the probability of getting an approved item, i.e. the probability of zero faulty measurements. We are usually not interested in calculating the probability of 1, 2, 3, etc faulty measurements. For quality improvements we will use the measurements themselves in order to understand the process behaviour.

 erase c1-c1000                                # Erases the worksheet.

 name c1 'M1' c2 'M2' c3 'M3' c4 'M4' c5 'M5'  # Names the measurements.

 name c6 'M6' c7 'M7' c8 'M8' c9 'M9' c10 'M10'

 # Names a number of columns that carry the information PASS/FAIL.

 name c11 'P/F M1' c12 'P/F M2' c13 'P/F M3' c14 'P/F M4' c15 'P/F M5'

 name c16 'P/F M6' c17 'P/F M7' c18 'P/F M8' c19 'P/F M9' c20 'P/F M10'

 name c21 'P/F tot'

 let k1 = 10000                      # Number of items to be simulated.

 random k1 c1-c4;                    # Simulates the normal values.

 normal 50 5.                        # Mu = 50, sigma = 5.

 let k2 = 1/0.9                      # Inverts the lambda-parameter.

 random k1 c5-c7;                    # Simulates gamma-values in c5-c7.

 gamma 2.5 k2.

 let k3 = 1/2.3                      # Inverts the lambda-parameter.

 random k1 c8-c10;                   # Gamma-values in c8-c10.

 gamma 15.1 k3.

 let c11 = (c1<41 or c1>62)          # Using a logical expression to 

 let c12 = (c2<41 or c2>62)          # create a '1' if outside spec.

 let c13 = (c3<41 or c3>62)

 let c14 = (c4<41 or c4>62)

 let c15 = (c5<0.5 or c5>5.4)        # Using a logical expression to

 let c16 = (c6<0.5 or c6>5.4)        # create a '1' if outside spec.

 let c17 = (c7<0.5 or c7>5.4)

 let c18 = (c8<3.1 or c8>9.2)        # Using a logical expression to

 let c19 = (c9<3.1 or c9>9.2)        # create a '1' if outside spec.

 let c20 = (c10<3.1 or c10>9.2)

 rsum c11-c20 c21                    # Sums all 1's across each item.

 let c21 = (c21>0)                   # A '1' if the item fails.

 let c22(1)  = sum(c11)/k1           # Calculates observed fault rate

 let c22(2)  = sum(c12)/k1           # for each one of the measurements.

 let c22(3)  = sum(c13)/k1           # The result is stored in c22.

 let c22(4)  = sum(c14)/k1

 let c22(5)  = sum(c15)/k1

 let c22(6)  = sum(c16)/k1

 let c22(7)  = sum(c17)/k1

 let c22(8)  = sum(c18)/k1

 let c22(9)  = sum(c19)/k1

 let c22(10) = sum(c20)/k1

 let c22(11) = sum(c21)/k1          # The total fault rate.

In order to compare the simulated results above, we need to calculate the corresponding theoretical result. Therefore we run two different macros – %NormArea and %GamArea:

1. Start a new session of Minitab

2. Run the macro %NormArea once

3. Enter the following data in c1: 50, 5, 41, 62

4. Rerun the macro %NormArea
(The probability inside the limits is 95.587%, this value is used below)

5. Clear the worksheet (e.g. by REST)

6. Run the macro %GamArea once

7. Enter the following data in c1: 2.5, 0.9, 0.5, 5.4

8. Rerun the macro %GamArea
(The probability inside the limits is 88.665%, this value is used below)

9. Enter the following data in c1: 15.1, 2.3, 3.1, 9.2

10. Rerun the macro %GamArea
(The probability inside the limits is 92.340%, this value is used below)

                                    # The total fault rate is (1 – an OK item)

 let c22(12) = 1 - 0.956**4 * 0.887**3 * 0.923**3

Conclusion.  The 11th and the 12th value of column c22 correspond quite well. The 11th value is the simulated result and the 12th value is the theoretical result.

Appendix I: calculating the distribution for example I

Calculating the exact probability distribution of Y, defined in the model above, is fairly compli​cated. It involves the rather advanced statistical and mathematical ideas called probability gene​rating func​tions. In order to do these calculations we use a computer program called Mathematica and the com​mands and results are included below for the sake of completeness only.

p1 = 0.03; p2 = 0.05; p3 = 0.08; p4 = 0.10; p5 = 0.12;

n1 = 15; n2 = 20; n3 = 25; n4 = 4; n5 = 3;

q1 = 1 - p1; q2 = 1 - p2; q3 = 1 - p3; q4 = 1 - p4; q5 = 1 - p5;

gen1 = (q1 + p1 x)^n1; gen2 = (q2 + p2 x)^n2;

gen3 = (q3 + p3 x)^n3; gen4 = (q4 + p4 x)^n4;

gen5 = (q5 + p5 x)^n5;

sumtot = gen1 *  gen2 * gen3 * gen4 * gen5

Expand[sumtot];

res = CoefficientList[%, x]

The commands calculate all possible 68 probabilities but only the first 15 are shown below as the rest are considered small (if there are 67 components etc, the number of possible number of faults are 68, i.e. including the '0'):

   0.012623   0.057359   0.128022   0.187090   0.201347   0.170172

   0.117623   0.068372   0.034109   0.014832   0.005690   0.001945

   0.000597   0.000166   0.000042

